Note 20th October 2019

There might be smarter ways of using OSPEX now than when this was written see:
https://hesperia.gsfc.nasa.gov/ssw/packages/spex/doc/ospex_explanation.htm
But the basic principles still hold for solar X-ray Spectroscopy, applied to RHESSI or other data….

Introduction to RHESSI Flare X-ray Spectroscopy

Iain G. Hannah

Department of Physics & Astronomy,
University of Glasgow, UK

3rd June 2009, Solaire SODAS, Glasgow, UK

Email: iain.hannah@glasgow.ac.uk

Science & Technology Facilities Council

Introduction & Motivation

- **Reuven Ramaty High Energy Solar Spectroscopic Imager**
  - RHESSI provides X-ray & γ-ray spectra and images of flares

- **We will focus on the X-ray spectrum of flares (3 to 100s keV)**
  - Provides direct information about the electrons accelerated in flares
    - Hard X-rays (HXR) , non thermal emission typically >10 keV
  - Information about the hot thermal emission >10 MK
    - Soft X-rays (SXR), thermal emission typically < 25 keV
  - Imaging covered by Eduard Kontar Friday morning

- **Outline:**
  - Flare X-ray emission & spectra
  - Introduction to RHESSI hardware
  - RHESSI data & software (RHESSI GUI & OSPEX)
  - Walk through examples
  - Further things......

"Typical" Flare X-ray Scenario

1. **Starts with a coronal energy release facilitated by magnetic reconnection**
2. **Outwards CME and electron beam**
   - Latter via Radio Type III or in-situ
3. **Downwards beam of accelerated electrons**
   - X-ray "thin-target" emission too faint
   - Also get microwaves
4. **Electron beam stopped in Chromosphere**
   - Bright X-ray "thick-target" footpoints
   - Stopped beam heats local plasma
5. **Hot material evaporated into coronal loop**
   - Initially observe this at hottest temperatures in SXR (>10 MK) then cools and seen in EUV



This is a gross simplification and things are generally more complicated. The exceptional cases are often the most interesting scientifically

“Typical” Light Curve & X-ray Spectra



Model photon spectra

Non-thermal (Hard) X-ray Emission

- **The bremsstrahlung free-free photon flux at the Earth $I(\varepsilon)$ is related to the source electron distribution $F(E)$ as:**

$$I(\varepsilon) = \frac{1}{4\pi R^2} \int_{\varepsilon}^{\infty} \int_{V} n(r) F(E, r) Q(\varepsilon, E) dE d^3 r$$

Photon Flux
ph s^{-1} cm^{-2} keV^{-1}

1 AU

Density of background plasma e- interacting with

e- distribution
e$^-$ cm^{-2} s^{-1} keV^{-1}

Bremsstrahlung cross-section

ε =Photon Energy
E =Electron Energy

- **Assume non-thermal emission from a power-law of electrons accelerated out of Maxwellian, spectral index δ above E_C** $$F(E) \propto E^{-\delta}$$

- **Total Number of accelerated electrons s^{-1} above E_C [keV]** $$N(E > E_C) = \int_{E_C}^{\infty} F(E) dE$$

- **Power in these electrons in erg s^{-1}** $$P(E > E_C) = \int_{E_C}^{\infty} F(E) E dE = 1.6 \times 10^{-9} \frac{\delta - 1}{\delta - 2} N E_C$$

- **Non-thermal Energy in erg** $$U_N(E > E_C) = P(E > E_C) \Delta t$$

Thin and Thick Target Hard X-rays

- **We normally investigate the emission within 2 limits**
  1. Thin target: energy losses not significant dU/dt≈0
     - Tenuous coronal emission => not observed
  2. Thick target: electrons stopped completely dU/dt=Coulomb rate
     - From dense chromosphere => bright footpoints emission

- **If take simpler non-relativistic form of Q(ε,E) find analytically** $I(\varepsilon) \propto \varepsilon^{-\gamma}$
  – Kramers or Non-relativistic Bethe-Heitler (see Brown 1971, Holman 2009)





*previous version didn't have the axes labelled log-log

Thick Target Hard X-rays

- **Get relationship between δ and γ and normalisation but not E_C**
  - E_C important as N and P depend on it $E_C^{-\delta}$
- **Also, from using more accurate Q(ε,E) and numerically solving get flattening at low energy**
    - Haug 1997 approximation to 3BN of Koch and Motz 1959
  - Thick target example since this is the emission we typically observed
    - Imaging helps confirm the source of the non-thermal emission





Which Function to Fit the Thick Target?

- **Broken power-law in photon space: `f_bpow.pro`**
  - Fast as only fitting simple photon model
  - But only getting vague feel for E_C and approx to turn over a low energies
  - Old approach, only used often due to its speed

- **Actual thick target model: `f_thick.pro or f_thick2.pro`**
  - Will recover all the non-thermal parameters we need to characterise the accelerated electrons
    - Implementation described in Holman ApJ 2004
  - But slow due to numerical integration
    - Electron distribution model then work out resulting photon spectrum
    - Recently faster version has been developed f_thick2.pro

Best to use f_thick or f_thick2

- Note that these are for a double power-law in e^- space
  - Occasionally useful in some large flare with harder higher energy component
- Trivial to make this single power-law: $\delta=\delta_1$, $E_B=E_{MAX}$

Thermal (Soft) X-ray Emission

- **Thermal emission mixture of thermal bremsstrahlung continuum and emission lines**
- **We use isothermal model continuum and spectral lines from CHIANTI**
  - **`f_vth.pro`**
  - Prescribed coronal setup from CHIANTI
- **Function of**
  - Temperature T in MK
    - Changes shape and spectral lines
  - Emission Measure EM=n^2V in cm^{-3}
    - Volume of emitting plasma V in cm^3
    - Density of emitting plasma n in cm^{-3}
    - Thermal spectrum linearly dependent on EM so just affects normalisation
  - Relative abundances
    - But we keep it =1



For more info see: Philips ApJ 655 2004

Spectral Forward Fitting

- **The count rate spectrum in the detectors is given by**

$$\mathbf{C} = \mathbf{B} + \mathbf{SRM\#I}$$

Count Spectrum — **Background** — **Spectrometer Response Matrix** — **Photon Spectrum**

- **SRM=Spectrometer Response Matrix**
  - Gives probability an incident photon creates a detected count
  - Not a diagonal matrix so the above is an Inverse Problem
- **Need to forward fit model to recover I(ε)**
  - Choose a photon model, convert to count model $\mathbf{C}_{\text{mod}} = \mathbf{SRM\#I}_{\text{mod}}$
  - Compare to observed counts spectrum,
  - Change model parameters to get closer match and repeat. $\mathbf{C}_{\text{obs}} \approx \mathbf{C}_{\text{mod}}$
  - Then can generate model photon spectrum
    - *Though highly model dependent* $\mathbf{I}_{\text{obs}} = \mathbf{C}_{\text{obs}} \dfrac{\mathbf{I}_{\text{mod}}}{\mathbf{C}_{\text{mod}}}$

This is what OSPEX does for us.........but first RHESSI to get C and SRM

RHESSI Spacecraft

- **Launched 5-Feb-2002, 1st light 12-Feb-2002**
- **In a Low Earth Orbit, inclination 38°**
  - 90 min orbit, 30 min night
- **9 pairs of collimating Grids with a segmented germanium detector behind each**
  - Covers 3 keV to 17 MeV
- **Other small detectors for very accurate pointing and roll info**
  - SAS, RAS, FSS
- **Spacecraft spinning with ~4sec period**
  - Time modulates solar signal with imaging information ...... See Kontar Friday
- **Continuously observing full Sun expect for**
  - Night-time or SAA periods in orbit
  - Offpointing for Crab



RHESSI's Detectors & Collimators

- **9 Segmented Germanium Detectors cooled to <100 K**
  - Time, energy and detector of each count recorded
    - Energy Resolution 1 keV FWHM <100 keV , 3 keV FWHM < 3 MeV
    - Time Resolution ~1μs
    - Unless decimation when a certain % ignored to save memory
  - Front segments stop X-rays <150 keV, Rear 150 keV to 20 MeV
    - No shielding so X-ray/γ-ray can get in from anywhere!









RHESSI's Attenuators/Shutters

- **To maintain dynamic range Aluminium shutters automatically come in to stop the excessive low energy photons**
  - A0 nothing in:
    - good> 3keV
  - A1 thin shutters in only:
    - good > 6 keV
  - A3 thin+thick shutters in:
    - good >12 keV





Pulse Pileup in RHESSI Detectors

- **If more than one detected count at similar time then electronics thinks it is 1 count at their total energy**
  - i.e. a 14 keV count instead of 2x 7 keV
  - Occurs with high count rate during bigger flares or before attenuators come in
    - High detector dead-time
    - $>10^3$ counts s^{-1} detector^{-1}
    - $>10^5$ counts s^{-1} detector^{-1}
      - correction not adequate as only fixes 2 pulses piling up
  - Model correction applied when generating count rate spectra
    - *See later........*



RHESSI Software & Data Access

- `SSW_INSTR = hessi hxt .......`

- **Within SSW need following environment variables setup:**
  - For local mirror of RHESSI data
    - `setenv HSI_DATA_ARCHIVE c:\hessi\data`
  - For your local store of RHESSI data
    - `setenv HSI_DATA_USER c:\hessi\data`
  - Or where ever you want to put the fits

- **In windows**
  - Edit this in ssw\site\setup\setup.hessi_env
  - If not there copy from ssw\hessi\setup\setup.hessi_env
- **In linux/mac -** Add lines to you .cshrc or .bashrc files

- ***VNC folk do nothing, mirror of RHESSI data connected to sirius.astro***

RHESSI Data Acces

- **Required hsi_200**fits and hsi_obssumm_200**fits on usb stick in \rhessi_data**
  - Copy to location specified in HSI_DATA_USER
  - This is data for Wednesday and Friday

- **For speed required hsi_spectrum*** and hsi_srm*** files have been pre generated and also on usb stick \rhessi_data**
  - Copy this to your current work dir (pwd or cd,'****)
  - VNC folk should find these in ~\rhessi_data

For other events

- **To automatically get the data when needed:**
  - `search_network,/enabled`
    - Can specify which server: `/ssl,/gsfc,/hedc`
  - Data will be downloaded to HSI_DATA_USER
    - or go to http://hesperia.gsfc.nasa.gov/hessidata/

2 types of RHESSI data

- **Level-0 data**
  – For each count record time, energy and detector
  – ~100b per 90 min quiet orbit or every ~100Mb during big flare
  – Needed for processing spectra and images

- **Quick look summary data**
  – Pre-processed into large energy bands per 4 sec over 6 detectors
    - Summary light curves
  – Flags determined as well -> flare, night, SAA, particle event etc
  – How to access
    - GUI->File->Observing Summary Data
    - Online Browser http://sprg.ssl.berkeley.edu/~tohban/browser/
    - PNGs in /hessidata/metadata/YYYY/MM/DD/

RHESSI online links

- **Main homepage – lots of links**
  - http://hesperia.gsfc.nasa.gov/hessi/
- **RHESSI Data & Software Centre – lots of info & tutorials**
  - http://hesperia.gsfc.nasa.gov/rhessidatacenter/
- **Tohban pages – lots of latest information & useful links**
  - http://sprg.ssl.berkeley.edu/~tohban/
- **RHESSI browser – online quick look data browser:**
  - http://sprg.ssl.berkeley.edu/~tohban/browser/
- **OSPEX tutorial and guide - Kim Tolbert's very useful pages**
  - http://hesperia.gsfc.nasa.gov/ssw/packages/spex/doc/ospex_explanation.htm
- **Earlier OSPEX tutorial – Brian Dennis August 2004**
  - http://hesperia.gsfc.nasa.gov/~dennis/OSPEX/The_Basics/index.htm
- **Plotman help - plotting interface GUI and OSPEX use**
  - http://hesperia.gsfc.nasa.gov/ssw/gen/idl/plotman/doc/plotman_help.htm

Ignore anything online about SPEX, the new better software is OSPEX.

RHESSI Software Overview

- **RHESSI GUI**
  - For quick look data, lightcurves, count spectra, images
  - At command line: **`hessi`**

- **Most things in the GUI can be done on the command line using the RHESSI objects**
  - Useful for repeating, automation, accessing advance features
  - We will focus on the GUI
  - GUI can save commands to script!

- **Spectrum fitting done using OSPEX**
  - Takes count rate spectrum and SRM produced but GUI and performs forward fitting



Analysing a RHESSI Flare Spectrum

1. **Finding a flare, what did RHESSI observe?**
   – Observing summary quick look light curves in GUI or online

2. **Generate count rate spectra & Spectrometer Response Matrix SRM fits file**
   – Count rate spectra for chosen energy binning, detectors for many time interval spanning the flare
      • Corrected for livetime, decimation, pulse pile-up, energy calibration
   – SRM gives matrix of probabilities that an incident photon at RHESSI produces a detected count

3. **Forward Fit a model photon spectrum to the count rate spectrum**
   – Load spectrum***.fits and SRM***.fits into OSPEX
   – Forward fit chosen model
   – Produce "observed" photon spectra and fitted parameters

Analysis Walk through

- **Going to go through 2 events in detail**
  - 6-Jan-2004 06:00 to 06:35
    - M6 flare multiple attenuator states
    - Detailed walk through
    1. Fit an A1 time interval with isothermal + thick target models

  - 26-Jun-2002 18:40 to 19:08
    - B8 flare with attenuator out A0
    - Quicker walk through
    1. Fit the peak time interval with isothermal + thick target models
    2. Fit multiple time intervals with isothermal + thick target models

1. Flare Finding/RHESSI Flare List

- **Either have an event/time range in mind or ....**
- **Use RHESSI Flare catalogue**
  - Accessible from GUI or type
    - `hsi_flarecat`
  - All found spikes in RHESSI data
    - http://hesperia.gsfc.nasa.gov/hessidata/dbase/hessi_flare_list.txt
    - Times, ID#, disk location, peak counts, flags
    - >48000 events but not all flares!
    - Actual flare: sflag=1 & x,y position on Sun
- **Multiple command line utilities**
  - http://hesperia.gsfc.nasa.gov/ssw/hessi/doc/guides/flare_list_utilities.htm




```
flares = hsi_read_flarelist()
help,flares,/str                        ; structure containing all flare info

flares = hsi_select_flare(start_time=['10-Jan-2003','12-Jan-2003'],$
         radial_dist_range=[300,900], peak_countrate_range=[1000,2000],$
         sflag=1,/structure)            ; structure containing flares within constraints
                                        ; without /structure just flare ID#
                                        ; /formatted instead gives ascii list
```

1. Summary Data/Observation Time

- **Observing summary data through GUI**
  - GUI plotting interface is plotman
    - Interact with plot
      - Left click+ drag=zoom region
      - Left click=max zoom out
      - Right click=mark x,y point
    - Output data and plot to file
- **Event of interest**
  - 06-Jan-2004 05:20:00 to 06:40:00
    - 06:00 to 06:35
- **To Finish “Set Obs Time And Close”**




```
ltc=hsi_obs_summary()
ltc->set, obs_time_interval='06-Jan-04 '+['05:20:00','06:40:00']
ltc->plotman,/saa,/flare,/night,/corrected,/ylog    ;no /plotman just in normal window

or
data=ltc->getdata(/corrected)
Time=ltc->getaxis(/ut)
utplot,anytim(time,/yoh),data.countrate[0]
```

1. Observing Summary Plot: B8 Flare

- **Use observing summary plots to check what RHESSI observed**
  - Prefer online browser as GOES, quicklook images etc all there



1. Observing Summary Plot: No Flares



Sinusoidal variation with 45min (or half orbit) period due to differing cosmic ray flux at geomagnetic latitudes.

Highest rate at highest latitudes.

Smaller higher energy spikes due to particle events. Separate RHESSI particle detector confirm this.

1. Observing Summary Plot : X1.8 Flare

- **Attenuators automatically come in to stop detector saturating:**
  - A0 then repeatedly to thin A1 (<25 keV) then thin+thick A3 (< 75 keV)





- **Corrected Count rate empirically corrected to A1 state**
  - Don’t use for science! As not properly calibrated

2. Generate Count Rate Spectra

- In RHESSI GUI, File->Retrieve/Process Data->Spectrum...



2. Energy & Time Binning

- **Energy binning**
  - Depends on size of flare, energy range of interest
    - Basically are there enough counts in all the energy bins of interest?
    - Do not do less than detector energy resolution
      - 1 keV FWHM <100 keV , 3 keV FWHM < 3 MeV
- **Either define manually or select pre-defined binning code**
  - Manually give energy edges: i.e. 3+findgen(100)/3.
  - Binning code 22 a decent option
    - 1/3 keV 3-15 keV, 1 keV 15-100 keV then 10keV 100-300 keV



- **Time binning -> just stick to 4seconds**
  - Shortest time independent of RHESSI rotation/data modulation
  - If want longer can sum together time intervals in OSPEX



2. Detectors and More Config Options

- **Normally use a number of summed detectors to improve SNR**
  - Although if want best energy resolution just use single detector, 3,4 or 8
  - Sum those with best energy response and behaviour
    - 1,3,4,6,8,9 and Front Segment only (<100 keV stopped in front)
    - Although occasional exceptions, i.e. Recently 6 has been noisy



- **Want Count flux and *NOT* Semi-calibrated**
  - Flux shows spectra corrected for detector dead-time and data gaps
  - Semi-calibrated: diagonal SRM only so low & high energy response wrong



- **Pileup Correction MAYBE, Decimation Correction YES, XY offset YES**
  - Fix to pileup when rate >10^3 count/s/det, not work <10^5 count/s/det
  - Take x,y position of flare from flare catalogue -> needed for calibration



2. Generate Spectra and SRM Files

- **Now ready to generate files -> "Write Output file...."**



  - Auto-name if you want
    - Output to current working directory
      - hsi_spectrum_20040106_060000.fits
      - hsi_srm_20040106_060000.fits



  - YES to Create SRM,
  - 0- Full Calculation...,
  - Write FITS file to start processing
    - Outputs to current working directory (**pwd**)
- **May take minutes to 10s minutes to process**
  - Takes longer for more energy, time bins and pile-up correction

- **To save time the needed spectra and SRM fits are on your usb stick**

2. Immediate Check

- **Can immediately plot time integrated spectrum (Plot Spectrum), spectrogram (Plot Spectrogram) or lightcurve (Plot Time Profile) to check it worked**





Attenuator changes very noticeable, as is night-time, short data gap as well

2. Spectra & SRM via Command Line

- **Can the do Write Script Option (think you might need/consider):**

```
obj = hsi_spectrum()
obj-> set, decimation_correct= 1
obj-> set, obs_time_interval= [' 6-Jan-2004 06:00:00', ' 6-Jan-2004 06:35:00']
obj-> set, pileup_correct= 0
obj-> set, seg_index_mask= [1,0,1,1,0,1,0,1,1,0,0,0,0,0,0,0,0,0]
obj-> set, sp_data_unit= 'Flux'
obj-> set, sp_energy_binning= 22
obj-> set, sp_semi_calibrated= 0
obj-> set, sp_time_interval= 4
obj-> set, sum_flag= 1
obj-> set, use_flare_xyoffset= 1
```

- **Additional lines to write out spectra and SRM fits files**

```
obj->filewrite, /fits, /buildsrm, srmfile='hsi_srm_20040106_060000.fits', $
     specfile = 'hsi_srm_20040106_060000.fits', all_simplify=0, /create
```

  - **srmfile** and **specfile** can be whatever name you want
  - Don't include them to get the automatic file naming

3. OSPEX: Spectral Forward Fitting

- **OSPEX is the object orientated version of SPEX (SPectral EXecutive)**
  - R. Schwartz & K. Tolbert
- **Within OSPEX we will**
  - Load the files into OSPEX
  - Choose background time(s)
  - Choose time(s) to fit
  - Choose model to fit
  - Do fit
    - Ideally not automated but adopt some strategy
  - Save/plot results
- **Run OSPEX: `o=ospex()`**
  - The name of the ospex object can be whatever you like, here it is “o”



3. Select Input Files

- **File->Select Input**
  - Browse and select the name of your Spectrum file
  - SRM file should be automatically found and rest of boxes filled out
  - Can Plot to check everything loaded properly
  - Close once done



3. Background Selection

- **Ideally pick time range before flare in SAME attenuator state**
  - Before as sharper defined due to impulsive rise profile of typical flares
- **If none suitable take night-time before or after flare**
  - For this event, A1 fit interval so night-time after background
- **File-> Select Background**
- **Change**





- **To Select background time**
  - Left click=start time,
  - Right click =end
  - Or manually Edit Selected Interval
- **06:32:00 to 06:33:20**
- **Accept and Close, then Close once done**



<u>Extras</u>

- **Function 0Poly as only one interval**
  - In some flares get before & after
  - So can extrapolate between them with chosen function
- **Also can choose different intervals for each energy band**
  - But only have night-time

3. Fitting the Spectrum

- File-> Select Fit Options and Do Fit.....

Time Interval(s) to Fit

Energy Range to Fit

Plot & Options



Choose Fit Time Interval

Fit Function & Options

Start the fitting interface

3. Choose Time Interval to Fit

- **To Select Fit time interval**
  - Left click=start time,
  - Right click =end
  - Or manually Edit Selected Interval
- **Double left click in defined time interval for more options**
  - Automatically defined multiple time intervals etc
- **Just want one time interval here**
  - 06:22:20 to 06:23:00
  - Remember time resolution here is 4 seconds
- **Accept and Close once done**



3. Perform the Forward fit

- Back in Fit Options choose "Do Fit" (Make sure Loop Mode is Manual)
- By default only Thermal Model so need to scroll to thick or thick2 and add



3. Fitting Strategy

DO NOT just vary all the parameters over all energies in a single fit

- Recap model is [EMx10^{49},T kev, Ratio=1,N, δ,E_B=E_{MAX},δ_2=20,E_C,E_{MAX}]
- Want to vary parameters [1,1,0,1,1,0,0,1,0]

Fitting Strategy to adopt for attenuator A1 interval

1. **Start by changing values and plotting until**
   a) Thermal parameters and non-thermal spectral index close to "actual"
   b) Make non-thermal N very small so thermal initially dominating
2. **Fit thermal only over thermal energy range**
   a) Parameters free [1,1,0,0,0,0,0,0,0] over 6 to 20 keV
      - 6 keV as A1, guessing 20 keV from looking at spectrum
3. **Fit non-thermal only over non-thermal energy range**
   a) Parameters free [1,1,0,1,1,0,0,1,0] over 20 to 300 keV
4. **Fit both components over whole energy range 6 to 300 keV**
   a) Parameters free [1,1,0,1,1,0,0,1,0]
5. **Then TWEAK.........**

3. Fitting Strategy

- This gets us to a reasonably, but not great, fit



Thermal fit only, poor but remember response < 6 keV not accurately know in A1



Non-Thermal fit only, good



Fit everything, not bad

3. When to Stop Fitting

- **This will TYPICALLY work in A1 but will often need tweaking...**
  - Check that the χ^2 is small
    - χ^2 is only for current energy range selected not all energies
  - Residuals ~0 with no obvious structure
  - Check that the fit makes physical sense:
    - Thermal dominates at low energies and T
    - Non-thermal spectral index 2 to 10, E_c >7 keV
  - Continuing to fit doesn't vary parameters too much
    - Typically get fit values probably can get 10% accuracy

- **In this event can't get much better than this, why?**
  - Bad background subtraction?
  - Does mean 6-20 keV fit not great, δ fine but E_{MAX}?



3. Plotting and Output

- **Plenty of flexible plotting options form GUI and output to png, ps, jpg, tiff**
  - Configure plot then Create plot

- **Can output setup and results in a variety of formats**
- **The most useful is writing out the fit results and script code to restore your OSPEX setup**
  - Write Script -> and Fit Results
  - Will only write out how to get to your final fit result not the process to get there.
  - Can also reload fit results from GUI

3. Final Fitted Spectrum

- **Remember that the observed count rate spectrum is converted to an observed photon rate spectrum so is FIT DEPENDENT**
  - Fit generally looks good in photon space but with forward fitting count space is more important





3. What to do with Fit Results

- **Show Fit Summary**
  - $EM=0.197\times10^{49}$ cm^{-3} ,T=1.71 keV (/0.086164 = 19.9 MK)
  - $N=1.23\times10^{35}$ electrons s^{-1} , $\delta=3.36$, $E_c=20.42$ keV
  - Results not as accurate as sigma of fit suggests
    - Just look at the spectrum!

- **Thermal Energy (we will take volume=2x10^{26} cm^3)**
  - `energy_th=3*sqrt(EM*Vol)*1.38x10`$^{-23}$`*TMK*10`^{6}`*10`^{7}
    - 1.6×10^{29} erg

- **Non-thermal Energy**
  - `energy_nn=1.6x10`$^{-9}$`*(`δ`-1)*N*E`$_C$`/(`δ`-2)*`Δ`t`
    - 2.8×10^{29} erg

3. Or to do it all from Command Line

```
o=ospex()
; if you want it fully automated
;o-> set, spex_fit_manual=0, spex_fit_reverse=0, spex_fit_start_method='previous_int'
;o-> set, spex_autoplot_enable=0, spex_fitcomp_plot_resid=0,spex_fit_progbar=0
o-> set, fit_function='vth+thick2'
o-> set, fit_comp_spectrum= ['full', '']
o-> set, fit_comp_model= ['chianti', '']
o-> set, spex_specfile='hsi_spectrum_20040106_060000.fits'
o-> set, spex_drmfile='hsi_srm_20040106_060000.fits'
o-> set, spex_bk_time_interval=['6-Jan-2004 06:32:00', '6-Jan-2004 06:33:20']
o-> set, spex_bk_order=0
o-> set, spex_fit_time_interval=['6-Jan-2004 06:22:20','6-Jan-2004 06:23:00']
o-> set, mcurvefit_itmax= 50
o-> set, spex_erange=[6,20]
o-> set, fit_comp_free=[1,1,0,0,0,0,0,0,0]
o-> set, fit_comp_param=[1e-3,1.5,1,1e-2,3.5,1000,20,40,1000]
o-> dofit
o-> set, spex_erange=[20,300]
o-> set, fit_comp_free=[0,0,0,1,1,0,0,1,0]
o-> dofit
o-> set, spex_erange=[6,300]
o-> set, fit_comp_free=[1,1,0,1,1,0,0,1,0]
o-> dofit
params=o-> get(/spex_summ_params)
o-> plot_spectrum,/show_fit,/bksub,spex_units='flux',/overlay_back,/show_err
o-> plot_spectrum,/show_fit,/bksub,/photon,spex_units='flux',/overlay_back,/show_err
```

Do not recommend running completely automated, use this code to just setup everything up and then manually tweak.

2^{nd} Walk Through: 26-Jun-2002

- **Already found the event**

```
timer='26-Jun-02 '+['18:15','19:30']
ltc=hsi_obs_summary(obs_time_interval=timer)
ltc-> plotman,/saa,/flare,/night,/corrected,/ylog
```

- **As microflare in A0 and low detector dead-time so no need for pileup and do 1/3 keV energy binning from 3 to 50 keV**

```
os = hsi_spectrum()
os-> set, decimation_correct= 1
os-> set, obs_time_interval= '26-Jun-02 '+['18:40','19:08']
os-> set, pileup_correct= 0
os-> set, seg_index_mask= [1,0,1,1,0,1,0,1,1,0,0,0,0,0,0,0,0,0]
os-> set, sp_data_unit= 'Flux'
; 1/3 keV from 3 to 50 keV
os-> set, sp_energy_binning=3.+findgen(142)/3.
os-> set, sp_semi_calibrated= 0
os-> set, sp_time_interval= 4
os-> set, sum_flag= 1
os-> set, use_flare_xyoffset= 1
os-> filewrite, /fits, /buildsrm,all_simplify=0
```

Load into OSPEX & Background Time

- **For convenience, already found background time**

```
o=ospex()
o-> set, fit_function='vth+thick2'
o-> set, fit_comp_spectrum= ['full', '']
o-> set, fit_comp_model= ['chianti', '']
o-> set, spex_specfile='hsi_spectrum_20020626_184000.fits'
o-> set, spex_drmfile='hsi_srm_20020626_184000.fits'
o-> set, spex_bk_time_interval='26-Jun-02 '+['18:43:20','18:45:40']
o-> set, spex_bk_order=0
```





Choose Multiple Time Intervals

```
o-> set, spex_fit_time_interval='26-Jun-02 '+['18:46:36','18:56:36']
```

- **Then do 10 subintervals gives us:**
  - N=10 then double left click region, select break into N Sub-regions





Fit 2nd Time Interval

Fitting Strategy for A0 event

1. **Start by changing values and plotting until**
    1. Again get thermal parameters and spectral index close, make normalisation of non-thermal component small
2. **Fit thermal only over thermal energy range**
    a) Parameters free [1,1,0,0,0,0,0,0,0] over 3 to 8 keV
        - 3 keV as A0, guessing 9 keV from looking at spectrum
3. **Fit non-thermal only over non-thermal energy range**
    a) Parameters free [0,0,0,1,1,0,0,1,0] over 9 to 20 keV
        - Above 20 keV mostly getting noise dominated
4. **Fit both components over whole energy range**
    a) Parameters free [1,1,0,1,1,0,0,0,0] over 3 to 20 keV
        - Don't vary Ec again has habit to tending to small values
5. **Then TWEAK.........**

Fit 2nd Time Interval via Command Line

```
o->set, spex_erange=[3,8]
o->set, fit_comp_free=[1,1,0,0,0,0,0,0,0]
o->set, fit_comp_param=[5e-4,1.2,1,1e-4,7,200,20,20,200]
o->dofit, spex_intervals_tofit=1
o->set, spex_erange=[9,20]
o->set, fit_comp_free=[0,0,0,1,1,0,0,1,0]
o->dofit, spex_intervals_tofit=1
o->set, spex_erange=[3,20]
o->set, fit_comp_free=[1,1,0,1,1,0,0,0,0]
o->dofit, spex_intervals_tofit=1

params=o->get(/spex_summ_params)
help,params,/str
    PARAMS          FLOAT     = Array[9, 10]
print, params
     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000
   0.00220200      1.10268      1.00000     0.423660      7.61497      200.000      20.0000      12.1488      200.000
     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000
     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000
     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000
     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000
     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000
     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000
     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000
     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000     0.000000
```

Plot the Resulting Spectra

```
o->plot_spectrum,/show_fit,/bksub,spex_units='flux',/overlay_back,/show_err
o->plot_spectrum,/show_fit,/bksub,/photon,spex_units='flux',/overlay_back,/show_err
```





- **print,energy_th,energy_nn**
  - 1x10^{28} erg, 5.8x10^{28} erg

Fitting Multiple Time Intervals

- **Automated fitting of multiple time intervals**
  - Needed options command line only just now
  - Though still tricky and should manually tweak once done

```
o-> set,spex_fit_start_method='previous_iter'
o-> set,spex_fit_init_params_only=1        ; this is command line only just now
o-> set,spex_fit_manual=0
o-> set, mcurvefit_itmax= 50
o-> set, spex_erange=[3,8]
o-> set, fit_comp_free=[1,1,0,0,0,0,0,0,0]
o-> set, fit_comp_param=[5e-4,1.2,1,1e-4,7,200,20,20,200]
o-> dofit,/all
o-> set, spex_erange=[9,20]
o-> set, fit_comp_free=[0,0,0,1,1,0,0,1,0]
o-> dofit,/all
o-> set, spex_erange=[3,20]
o-> set, fit_comp_free=[1,1,0,1,1,0,0,0,0]
o-> dofit,/all
```

Plotting Multiple Parameters

- **For multiple time intervals OSPEX can plot the params vs time or each other**
  - File-> Plot Fit Results....
  - Expand vth or thick2,
- **To get Param vs Time**
  - Choose variable & Plot
    - Deselect Free/Fixed Bar
- **To get Param vs Param**
  - Choose 2 variable & Plot 1st vs 2nd
- **Of course as already shown can get all the data and fit parameters out of the OSPEX object and plot manually**





University of Glasgow

Multi-Panel Options in Plotman

- **Window_Control -> Multiple Panel Options..**





Extras/Advance Things

- **Albedo correction**
  - Photospheric Compton backscatter of the bremsstrahlung X-rays can significantly modify the spectrum in the RHESSI energy range
  - Correction implemented in RHESSI SSW by Kontar

- **Imaging Spectroscopy**
  - Instead of doing full solar disk spectroscopy get the spectrum from different parts of the flare
    - Requires you to make lots of images in many energy bands

- **Inversion to get electron spectrum from observed counts**
  - Instead of getting a photon one and interpreting what the electrons are doing
  - Various algorithms in RHESSI software to try and do this
    - i.e. Regularised Inversion technique implemented by Kontar et al.

Conclusions

- **RHESSI is a great instrument that lets us investigate accelerated electrons and hot plasma in flares**
- **RHESSI GUI & OSPEX makes the analysis relatively easy**
  - Just press the buttons !!!
- **But proper spectrum fitting is still a bit of an art**
  - Need to understand the physics of the emission
  - Need to understand the instrument
- **This is particularly true for exceptional/interesting events**
  - Shown here were fairly straightforward flares
  - Is this a a unique event or have I just done the analysis wrongly?
- **Do not be scared and there are 100,000s of flares to play about with**
- **And you can also do imaging........ Kontar, Friday morning**